

05.a MiM EEPROM

Na ploči RTC modula nalazi se i EEPROM 24LC32 kapaciteta 4kB (4096 bajta) koji se može koristiti za trajno smeštanje podataka.

```
// *****
// eeprom_primer
// *****
// Koristi eeprom 24lc32 gde upisuje jedan podatak (val) na
// adresu u memoriji (address).Zatim cita isti podatak iz
// eeprom-a i ispisuje na Serial Monitor: adresu sa koje je
// procitan i njegovu vrednost.
// Napomena: 24lc32 ima 4096 adresa, odnosno 12 bita za
// adresiranje.
// *****


#include "Wire.h"
#define EEPROM_I2C_ADDRESS 0x50 // I2C adresa za 24LC32 je 1010A2A1A0

void setup()
{
    Wire.begin(); // Pocinje I2C prenos
    Serial.begin(9600); // Za Serial Monitor
    int address = 0; // Adresa u EEPROM-u za upis podatka
    byte val = 234; // Podatak koji se upisuje na tu adresu
    writeEEPROM(address, val); // Upis podatka u EEPROM
    byte readVal = readEEPROM(address); // Citanje podatka iz EEPROM-a
    Serial.print("Sa adrese iz memorije: ");
    Serial.println(address);
    Serial.print("Procitan je podatak: ");
    Serial.println(readVal);
}


void loop(){}

void writeEEPROM(int address, byte val) //Upis u EEPROM
{
    Wire.beginTransmission(EEPROM_I2C_ADDRESS); //Adresa 24lc32
    Wire.write((int)(address >> 8)); // MSB deo adrese za upis
    Wire.write((int)(address & 0xFF)); //LSB deo adrese za upis
    Wire.write(val); // Upis vrednosti
    Wire.endTransmission(); // Kraj I2C prenosa
    delay(5); // Da memorija završi upis
}

byte readEEPROM(int address) // Funkcija za citanje EEPROM-a
{
    byte rData = 0x00; // Inicijalizacija promenljive
    Wire.beginTransmission(EEPROM_I2C_ADDRESS); //Adresa 24lc32
    Wire.write((int)(address >> 8)); //MSB deo adrese za upis
    Wire.write((int)(address & 0xFF)); //LSB deo adrese za upis
    Wire.endTransmission(); // Kraj I2C prenosa
    Wire.requestFrom(EEPROM_I2C_ADDRESS, 1);
    rData = Wire.read();
    return rData;
}
```




Slika 1: Format za upis podatka u 24LC32 (za 4096 adresa potrebno je 12 bita, pa adresni deo koristi 2 bajta)



Slika 2: EEPROM na modulu je čip sa oznakom U3


FIGURE 5-1: CONTROL BYTE FORMAT



Slika 3: Adresiranje EEPROM čipa

Zadatak:

- * Povezati RTC modul sa Arduino pločomi testiratinavedeni eeprom_primer
- * Napisati program koji pomoću senzora DS18B20 meri temperaturu u zadanom trenutku vremena (npr. u 6 sati) i rezultat upisuje u EEPROM
- * Ako bi se merenje i upis podatka u EEPROM obavljalo svakog punog sata izračunati za koje vreme bi se popunio ceo prostor u 24LC32.



Slika 4: RTC modul i senzor temperature DS18B20 (spaja se u 3 otvora na pločici gore levo)